
HOWTO: Display Outlook Folder Names

Article ID: Q174141

The information in this article applies to: 

· Microsoft Visual Basic Control Creation, Learning, Professional, and Enterprise Editions for Windows, versions 5.0, 6.0 

SUMMARY

The following code sample demonstrates how to print the names of all folders grouped under a specified Outlook folder. The sample assumes that the Microsoft Outlook mail client is installed. 

MORE INFORMATION

The code below uses a recursive routine to iterate through a mail folder to produce a list of all its sub-folders in the Immediate Window. 

Step-by-Step Example

1. Install Microsoft Outlook if it is not already installed. 

2. Open a Standard EXE project in Microsoft Visual Basic. 

3. Add a reference to the Microsoft Outlook 8.0 Object Library 

  (msoutl8.olb) using the Project item on the References menu.

4. Set the Startup Object to "Sub Main" from the Project Properties dialog. 

5. Add a Standard Module (.BAS) file to the project. 

6. Insert the following code into the module: (Modify the FOLDER_TO_OPEN 

   constant in the code below as appropriate and Execute.)

      Option Explicit

      Private Sub Main()

        Dim olMAPI As Outlook.NameSpace

        Dim Folder As Outlook.MAPIFolder

        '// Modify as appropriate.

        Const FOLDER_TO_OPEN = "Mailbox - John Doe"

        Set olMAPI = GetObject("",

   "Outlook.application").GetNamespace("MAPI")

        Call PrintFolderNames(olMAPI.Folders(FOLDER_TO_OPEN), "-

   >")

        Set olMAPI = Nothing

      End Sub

      Sub PrintFolderNames(tempfolder As Outlook.MAPIFolder, a$)

        Dim i As Integer

        If tempfolder.Folders.Count Then

          Debug.Print a$ & " " & tempfolder.Name

          For i = 1 To tempfolder.Folders.Count

            Call PrintFolderNames(tempfolder.Folders(i), a$ & "-

   >")

          Next i

        Else

          Debug.Print a$ & " " & tempfolder.Name

        End If

      End Sub

7. Run the project. 

NOTE: The sample code in this article works only when the FOLDER_TO_OPEN is set to a folder you can open, such as your own folder or the public folder. Otherwise a runtime error is generated. 

HOWTO: Speed Up Data Access by Using BeginTrans & CommitTrans

Article ID: Q146908

The information in this article applies to: 

· Microsoft Visual Basic for Windows Learning, Professional, and Enterprise Editions, versions 5.0, 6.0 

· Professional and Enterprise Editions of Microsoft Visual Basic, 16-bit, for Windows, version 4.0 

SUMMARY

You can speed up database operations in a Microsoft Access database by using transactions. A transaction starts with a BeginTrans statement and ends with a CommitTrans or Rollback statement. However, for reasons relating to stack usage it is recommended that these code regions are kept as small as possible. This ensures optimal performance and reliability." For more information as to when to use transactions, please see the following article in the Microsoft Knowledge Base: 

   ARTICLE-ID: Q145757
   TITLE     : Ideas to Consider When Using Transactions

The sample program below is over 17 times faster when using BeginTrans/CommitTrans. Performance may vary on different computers. 

MORE INFORMATION

You can tune the performance of Visual Basic by using transactions for operations that update data. A transaction is a series of operations that must execute as a whole or not at all. You mark the beginning of a transaction with the BeginTrans statement. You use the Rollback or CommitTrans statement to end a transaction. 

You can usually increase the record updates per second (throughput) of an application by placing operations that update data within an Access Basic transaction. 

Because Visual Basic locks data pages used in a transaction until the transaction ends, using transactions prevents access to those data pages by other users while the transaction is pending. If you use transactions in a multi-user environment, try to find a balance between data throughput and data access. 

If database operations are not within a transaction, every Update method causes a disk write. 

Transactions are very fast because they are written to a buffer in memory instead of to disk. CommitTrans writes the changes in the transaction buffer to disk. 

Robust error trapping is important when using transactions to avoid losing writes if the program gets an error in the middle of a transaction. 

For more performance tuning tips for data access in Microsoft Visual Basic version 4.0, see the PERFORM.TXT file. 

Step-by-Step Example

1. Start a new project in Visual Basic. Form1 is created by default. 

2. Add the following to the Form Load event code: 

      Private Sub Form_Load ()

         Dim Starttime, Endtime

         Dim db As Database

         Dim t As RecordSet

         Dim i As Integer

         Dim tempName As String

         Dim temphone As String

         Set db = Workspace(0).OpenDatabase("c:\vb\BIBLIO.MDB") ' Uses a

          ' copy of BIBLIO.MDB.

         Set t = db.OpenRecordSet("Publishers", dbOpenTable)

         Starttime = Now

         'BeginTrans  ' Add this and CommitTrans (below) for greater speed.

         For i = 1 To 100

            tempName = "testname" & Str$(i) ' Make an arbitrary unique

                                            '  string.

            tempPhone = Str$(i)             ' Make arbitrary number.

            t.AddNew ' AddNew clears copy buffer to prepare for new record.

            t!PubID = 30 + i  ' Set primary key to unique value.

            t!Name = tempName  ' Set Name field to unique value.

            t!Telephone = tempPhone  ' Set Telephone field to unique value.

            t.Update   ' Write the record to disk or to transaction buffer.

         Next i

         'CommitTrans  ' Add this and BeginTrans (above) for greater speed.

         Endtime = Now

         MsgBox "Time required= " & Format(Endtime - Starttime, "hh:mm:ss")

         t.Close

         db.Close

         End

      End Sub

   The above code adds 100 new records to the BIBLIO.MDB database file.

   Add the records to a copy of BIBLIO.MDB instead of to the original.

3. Start the program (or press the F5 key). A message box reports the time 

   required to add 100 new records. Close the form to end the program.

If you do not use the BeginTrans and CommitTrans statements, this program reports 17 seconds to add 100 records on a 486/66 PC. When you add BeginTrans and CommitTrans as shown in the program comments above, the program takes less than 1 second on that same computer. Performance may vary on different computers. 

· Microsoft Visual Basic, version 4.0, "Professional Features Book 1," Page 204 

· "Microsoft Developer Network News" newspaper, January 1994, Volume 3, Number 1, published by Microsoft Corporation. 

HOWTO: Use CoCreateGUID API to Generate a GUID with VB

Article ID: Q176790

The information in this article applies to: 

· Microsoft Visual Basic Control Creation, Learning, Professional, and Enterprise Editions for Windows, versions 5.0, 6.0 

· Microsoft Visual Basic Standard, Professional, and Enterprise Editions, 32-bit only, for Windows, version 4.0 

SUMMARY

As a programmer, you may need to generate GUIDs (Globally Unique Identifiers) for various purposes. This article describes how to generate a GUID in Visual Basic using the CoCreateGuid API. 

NOTE: The code in this article is not intended and cannot be used to create or change a GUID automatically generated by Visual Basic for custom ActiveX components. GUIDs automatically generated by Visual Basic cannot be altered. 

MORE INFORMATION

The code below can be used to create a GUID in Visual Basic. The code calls the CoCreateGuid API found in OLE32.DLL on both Windows 95, Windows 98, and Windows NT. In order to call the API correctly, a variable of type GUID must be passed. This code creates a custom type, named GUID, with four parts that represent the individual parts separated by dashes that you would see when viewing a CLSID or GUID in the system registry. This code simply returns a GUID; however, it can be modified to add the dashes if desired: 

Step By Step Example

1. Add a standard module to a new Visual Basic project. Form1 is created by 

   default.

2. Paste the code below into the code module: 

      Option Explicit

      Type GUID

         Data1 As Long

         Data2 As Integer

         Data3 As Integer

         Data4(7) As Byte

      End Type

      Declare Function CoCreateGuid Lib "OLE32.DLL" (pGuid As GUID) _

              As Long

      Global Const S_OK = 0 ' return value from CoCreateGuid

      Function GetGUID() As String

        Dim lResult As Long

        Dim lguid As GUID

        Dim MyguidString As String

        Dim MyGuidString1 As String

        Dim MyGuidString2 As String

        Dim MyGuidString3 As String

        Dim DataLen As Integer

        Dim StringLen As Integer

        Dim i%

        On Error GoTo error_olemsg

        lResult = CoCreateGuid(lguid)

        If lResult = S_OK Then

           MyGuidString1 = Hex$(lguid.Data1)

           StringLen = Len(MyGuidString1)

           DataLen = Len(lguid.Data1)

           MyGuidString1 = LeadingZeros(2 * DataLen, StringLen) _

              & MyGuidString1 'First 4 bytes (8 hex digits)

           MyGuidString2 = Hex$(lguid.Data2)

           StringLen = Len(MyGuidString2)

           DataLen = Len(lguid.Data2)

           MyGuidString2 = LeadingZeros(2 * DataLen, StringLen) _

              & Trim$(MyGuidString2) 'Next 2 bytes (4 hex digits)

           MyGuidString3 = Hex$(lguid.Data3)

           StringLen = Len(MyGuidString3)

           DataLen = Len(lguid.Data3)

           MyGuidString3 = LeadingZeros(2 * DataLen, StringLen) _

              & Trim$(MyGuidString3) 'Next 2 bytes (4 hex digits)

           GetGUID = _

              MyGuidString1 & MyGuidString2 & MyGuidString3

           For i% = 0 To 7

              MyguidString = MyguidString & _

                   Format$(Hex$(lguid.Data4(i%)), "00")

           Next i%

           'MyGuidString contains last 8 bytes of Guid (16 hex digits)

           GetGUID = GetGUID & MyguidString

        Else

           GetGUID = "00000000" ' return zeros if function unsuccessful

        End If

        Exit Function

      error_olemsg:

         MsgBox "Error " & Str(Err) & ": " & Error$(Err)

         GetGUID = "00000000"

         Exit Function

      End Function

      Function LeadingZeros(ExpectedLen As Integer, ActualLen As Integer) _

          As String

         LeadingZeros = String$(ExpectedLen - ActualLen, "0")

      End Function

3. Add a Command Button to the form, and add the following code to the 

   form:

      Private Sub Command1_Click()

           MsgBox GetGuid

      End Sub

4. Press F5 to run the project, and click the Command Button. 

RESULT: A GUID is generated and shown within a MessageBox. 

